
آشنايي با ساختار و اجزاي برنامه در محيط زبان برنامه​سازي پاسكال – 1

12-1- ساختمان عمومي برنامه

هر برنامه در زبان دلفي، از يك تكه متن اصلي برنامه و واحدهاي آن تشكيل مي​شود (شكل 12-1). 

براي ديدن ساختار كلي برنامه وارد منوي Project شده، View Source را باز كنيد. پنجره باز شده، ساختار كلي برنامه را نشان مي​دهد. اين ساختار، مشابه چنين چيزي است:

Program Project1;

uses

  Forms,

  Unit1 in 'Unit1.pas' {Form1};

{$R *.res}

Begin
  Application.Initialize;

  Application.CreateForm(TForm1, Form1);

  Application.Run;

End.
برنامه با دستور Program شروع مي​شود. اين دستور يك نام به برنامه نسبت مي​دهد. در بخش Uses واحدهاي بكار رفته در برنامه، نام برده مي​شود. دستورات اجرايي برنامه نيز در بدنه برنامه كه مابين Begin و End است، قرار مي​گيرند. در بدنه، ابتدا يك آماده​سازي اوليه (Initialize) صورت مي​گيرد و سپس نام فرمهاي برنامه تك به تك صدا زده مي​شود.
برنامه​هاي كه تا به حال در محيط دلفي نوشتيم، در واحدي با نامUnit1 ذخيره شده بود. اين واحد در فايلي با نام Unit1.pas قرار دارد. واحد، يك تكه از برنامه است، بطوريكه يك برنامه مي​تواند از چندين واحد تشكيل گردد. در فصل شانزدهم با ساختار و نحو واحدها بيشتر آشنا مي​شويم.

آنچه گفتيم ساختمان عمومي يك برنامه معمولي است.

12-2- محيط كنسول

برنامه​هايي كه تا به حال ايجاد كرديم، همگي از نوع يك محيط استاندارد ويندوز بوده​اند. اين نوع برنامه​ها از استاندارهاي خاصي استفاده مي​كنند. اما همه برنامه​ها از قبل چنين نبودند. برنامه​ها معمولا در محيطي نوشته مي​شدند كه هيچ رابط كاربر مستقلي نداشت، و خود برنامه​نويس مي​بايست فرم و اجزاي ديگر برنامه را توليد مي​كرد. البته در محيط دلفي نيز مي​توان برنامه​ها را در محيطي نوشت كه در آن هيچ چيز اوليه​اي وجود نداشته باشد، درست مانند محيطهاي برنامه​نويسي قديمي.
اين محيط، محيط كنسول نام دارد. براي ورود به اين محيط، از منوي File وارد منوي فرعي New شويد و از آنجا Other را انتخاب كنيد. سپس در پنجره باز شده، ConsoleApplication را انتخاب نماييد. با اين كار يك نوع برنامه كنسول ايجاد مي​شود. همانطور كه ديديد در پنجره New چندين نوع برنامه مي​توان انتخاب و ايجاد نمود. NewApplication كه ما در برنامه​هاي قبلي ايجاد مي​كرديم، تنها يكي از حالتهاي مختلفي است كه در برنامه مي​تواند ايجاد شود. 

با ايجاد يك برنامه كنسول مي​بينيد كه اين محيط، محيطي بدون فرم و واحد است و خود مستقيما يك برنامه است (شكل 12-2). اين محيط مشابه محيط پاسكال معمولي است و برنامه​نويسي در اين محيط تقريبا مشابه با برنامه​نويسي در پاسكال و در محيط سيستم عامل DOS است. 

ساختار كلي يك برنامه در اين محيط به صورت زير است:

Program Project1;

{$APPTYPE CONSOLE}

uses

  SysUtils;

Begin
  { TODO -oUser -cConsole Main : Insert code here }

End.
چون در محيط كنسول از ابزارهاي ورودي و خروجي موجود در محيط دلفي خبري نيست، بنابراين براي اعمال ورودي و خروجي از يك سري توابع استفاده مي​شود. 

چند نمونه از اين توابع عبارتند از:

Read , Readln : براي خواندن ورودي

Write , Writeln : براي چاپ خروجي 

با يك مثال، چگونگي ساخت يك برنامه در محيط كنسول را مشاهده مي​كنيم. هدف از نوشتن اين برنامه در محيط كنسول، آشنايي با اين محيط است. ضمن اينكه با مقايسه اين نوع برنامه​ها با برنامه​هايي كه در محيط تحت ويندوز نوشته مي​شوند، خواهيد فهميد كه برنامه​نويسي در يك محيط تحت ويندوز، تا چه اندازه آسانتر و زيباتر از يك محيط كنسول است.
مثال 12-1

برنامه جديدي از نوع كنسول در دلفي باز كنيد و سپس برنامه​اي بنويسيد كه فاكتوريل يك عدد را محاسبه كند.

Var

  n,

  i,

  fact : integer ;

 Begin
    writeln('please enter a number :') ;

    Readln(n) ;

    Fact := 1 ;

    for I := 2 to n do

        Fact := fact * I ;

    Writeln(fact) ;

    Readln ;

End.
دستور readln در محيط كنسول يك ورودي را دريافت مي‌كند. دستور writlen نيز يك خروجي را چاپ مي‌كند. دستور Readln بدون پارامتر در انتهاي برنامه، منجر به انتظار فشار دادن كليد enter مي‌شود.

با اجراي برنامه خواهيد ديد كه محيط اجراي برنامه چنين برنامه​هاي مانند محيط DOS است. اين برنامه و تمام برنامه​هاي كنسولي،‌ در  محيط DOS   اجرا مي‌شود (شكل 12-3).

12-3- نام​گذاري شناسه
​ها
همانطور كه در برنامه​هاي مختلف ديديد، براي متغيرها و ثابتها و .. در برنامه اسمي انتخاب مي‌كرديم. مثلا براي متغير شمارنده از نام I استفاده كرديم و توسط اين نام، آن متغير را مشخص كرديم. به اين نامها شناسه گويند. اما در يك برنامه​نويسي مجاز به انتخاب هر نوع اسمي نيستيم، بلكه بايد از قانون انتخاب نام آن زبان پيروي كنيم. در زبان پاسكال نيز قانوني براي انتخاب نام وجود دارد كه به شرح زير است:
· اسامي و شناسه​ها در زبان پاسكال از حروف A ... Z و a ... z و ارقام 0 ... 9 نيز كاراكتر "_"
 تشكيل مي​شوند. مثلا:
Sum , sum_value , point2
· اسامي نمي​توانند با ارقام شروع شوند. مثلا چنين شناسه​اي مجاز نيست:  2status   
· حروف كوچك و بزرگ در اسامي زبان پاسكال با هم متفاوت نيستند. مثلا اسامي زير همگي يكي هستند:
 CalculateValue

 calculateValue

calculatevalue

               CALCULATEVALUE
· طول يك شناسه مي​تواند تا 255 حرف باشد. مثلا:
Sum_of_first_list
· برخي از كلمات دستورات مانند If و Then و Begin و ... به عنوان كلمات ذخيره شده
 محسوب مي​شوند و نمي​توان از آنها براي نامگذاري و شناسه استفاده نمود. ليست اين كلمات را مي​توانيد در راهنماي دلفي مشاهده كنيد.
12-3-1- نكاتي مهم در نام​گذاري

· انتخاب نام مناسب اهميت بسياري در برنامه​نويسي دارد. مثلا استفاده از نام sum براي يك متغير، مشخص كننده آن است كه مجموع يك محاسبه در آن قرار مي​گيرد. اما نام aa يا t يا bbb (چنانچه توسط بسياري از برنامه​نويسان شلخته متداول است!) معلوم نمي​كند كه كار اين متغير چيست و براي چه چيزي استفاده مي​شود. حتي اگر يك متغير براي كارهاي مختلفي مورد استفاده قرار مي​گيرد، بايد نام آن به گونه​اي انتخاب شود كه اين را برساند. مثلاTemp 
· تنها وقتي به عنوان يك شمارنده (نظير يك شمارنده حلقه) يا يك پرچم
 علامت (كمي جلوتر اين نوع متغير را معرفي مي​كنيم) از متغيرها استفاده مي​كنيم، مي​توانيم از نامهايي مانند I , j , k و نظاير آن استفاده كنيم. در غير اين صورت، نام بايد مشخص كننده مفهوم و كاربرد متغير يا تابع مربوطه باشد.
· براي تفكيك بندهاي اسامي بلند يا از "_" استفاده مي​كنيم و يا در تايپ نامها، حروف ابتداي هر بند را بزرگ مي​نويسيم. مثلا :    First_command_code يا FirstValueAvarage 

12-4- متغيرها در زبان پاسكال

براي نگه داشتن اطلاعات از انواع مختلف در برنامه از متغيرهاي مختلفي استفاده مي​شود. در فصول گذشته ما تنها با متغير عدد صحيح يا integer آشنا شديم. در اينجا با انواع ديگري از متغيرها آشنا مي​شويم.
انواع متغيرهاي عددي- متغيرهاي عددي بر دو نوع اصلي تقسيم مي​شوند:
الف- اعداد صحيح (integer type)

ب- اعداد اعشاري (real type يا floating points)

يك متغير اعشاري مي​تواند يك عدد صحيح را در خود ذخيره كند. اما يك متغير صحيح نمي‌تواند يك عدد اعشاري را نگهداري كند. در عوض عمليات بر متغيرهاي اعداد صحيح دهها بار سريعتر است.

حجم و ظرفيت يك متغير: آب را مي​توان هم در ليوان و هم در يك بطري كوچك و هم در يك بشكه نگهداري كرد. اما چيزي كه در اين ميان متفاوت است، حجم آبي است كه در هر يك از اينها نگهداري مي​شود. يك متغير نيز چنين وضعيتي دارد. مثلا نوع متغير عددي صحيح وجود دارد كه تنها اعداد 0 تا 255 را در خود نگهداري مي​كند و انواع متغيرهاي ديگري وجود دارند كه اعداد بيشتري را در خود ذخيره مي​كنند. بستگي به اندازه متغيرمان نوع آن را تعيين مي​كنيم.
12-4-1- انواع متغيرهاي عددي صحيح

در جدول 1-12، انواع متغيرهاي عددي صحيح با ذكر محدوده عددي و اندازه بيتي​شان آمده​اند:

	اندازه و فرمت بيتي
	محدوده عددي
	نوع متغيرها

	signed 32-bit

	–2147483648..2147483647

	integer

	Unsigned 32-bit

	0..4294967295

	Cardinal

	Signed 8-bit

	–128..127

	Shortint

	signed 16-bit

	–32768..32767

	Smallint

	signed 32-bit

	–2147483648..2147483647

	longint

	signed 64-bit

	–2^63..2^63–1

	Int64

	unsigned 8-bit

	0..255

	Byte

	Unsigned 16-bit

	0..65535

	Word

	Unsigned 32-bit

	0..4294967295

	longword


12-4-2- متغيرهاي اعشاري

هر متغير اعشاري دربرگيرنده سه بخش است:

يك بيت براي علامت (مثبت يا منفي)

يك بخش عدد

يك بخش نما يا توان

تعداد بيت عدد و نما مشخص كننده محدوده​اي است كه مي​تواند نمايش داده شود. در جدول 2-12 انواع متغيرهاي عددي اعشاري با ذكر محدوده عددي، اندازه بيتي و رقم معني​دارشان آمده است.
	اندازه بيتي
	تعداد ارقام معني​دار
	محدوده عددي
	نوع متغيرها

	6
	11-12
	2.9 x 10^–39 .. 1.7 x 10^38

	Real48

	4
	7-8
	1.5 x 10^–45 .. 3.4 x 10^38

	Single

	8
	15-16
	5.0 x 10^–324 .. 1.7 x 10^308

	Double

	10
	19-20
	3.6 x 10^–4951 .. 1.1 x 10^4932

	ExtEnded

	8
	19-20
	–2^63+1 .. 2^63 –1

	Comp

	8
	19-20
	–922337203685477.5808..
922337203685477.5807

	Currency

	8
	15-16
	5.0 x 10^–324 .. 1.7 x 10^308

	Real

محاسبه اعداد در دو حالت اعشاري و صحيح انجام مي​شود. وقتي در يك عبارت تمام اعداد صحيح باشند، محاسبه صحيح و وقتي حداقل يك عدد يا عملگر اعشاري (نظير تقسيم) وجود داشته باشد، محاسبه اعشاري انجام مي​شود. مثلا 2+3/2 يك محاسبه اعشاري است و حاصل آن نيز يك عدد اعشاري يعني 3/4 است و 4+6 يك محاسبه صحيح و نتيجه آن نيز صحيح است. حتي 4+0/2 نيز يك محاسبه اعشاري و حاصل آن 0/6 يك عدد اعشاري است.

12-5- انتصاب اعداد

در عمل انتصاب اعداد، عبارت سمت راست محاسبه شده و نتيجه در متغير سمت چپ قرار مي​گيرد. در صورتيكه متغير سمت چپ صحيح باشد، سمت راست تنها مي​تواند يك عدد صحيح باشد. اما در صورتيكه متغير سمت چپ يك اعشاري باشد، سمت راست هم مي​تواند صحيح و هم اعشاري باشد.
انتصاب اعشاري به صحيح:  با استفاده از تابع جزء صحيح (براكت) يا Trunc مي​توان بخش صحيح يك عدد اعشاري را در يك متغير صحيح قرار داد.

I: integer ; 

r : real ;

…..

I := trunc ( r ) ;
مسئله تقسيم اعداد صحيح- مي​دانيم كه حاصل تقسيم معمولي يك عدد صحيح به عدد صحيح ديگر، يك عدد اعشاري است. براي انجام عمل تقسيم صحيح و محاسبه باقي​مانده آن از دو تابع DIV و  MOD استفاده مي‌كنيم.

تمرين

 برنامه​اي بنويسيد كه دو عدد دريافت كند و بخش​پذيري عدد اول را به عدد دوم كنترل كرده و اعلام كند. سپس اين برنامه را به صورتي تغيير دهيد كه يك عدد را دريافت كرده و كليه اعداد بخش​​پذير بر آن را در يك Memo نشان دهد.
سپس بدون استفاده از توابع mod و div، برنامه كنترل بخش​پذيري اعداد به هم را بنويسيد.

12-6- برخي از توابع عددي

براي كار با اعداد، در زبان دلفي توابع زيادي وجود دارند. برخي از اين توابع به شرح زير هستند:

ABS – قدرمطلق عدد را برمي​گرداند.
Int – بخش صحيح يك عدد اعشاري را برمي​گرداند.
Log10 - لگاريتم عدد را در مبناي ده برمي​گرداند.
Log2 – لگاريتم عدد را در مبناي دو برمي​گرداند.
Sqr – مربع يك عدد را برمي​گرداند.
Sqrt – ريشه دوم عدد را برمي​گرداند.
مثال 12-2

همان​طور كه گفته شد، توابع عددي زيادي وجود دارند. با نوشتن يك برنامه كوچك، مي​توان به عملكرد هر يك از اين توابع پي برد. برنامه زير يك عدد اعشاري را از ورودي مي​گيرد و عملكرد برخي از توابع اعشاري را روي آن نشان مي​دهد. با اجراي اين برنامه و دادن مقادير مختلف به متغير r، مي​توان عمل هر يك از توابع بكار گرفته شده  روي متغير r را مشاهده كرد. فرم اين برنامه در شكل 12-4 نشان داده شده است.

Procedure TForm1.Button1Click(SEnder: TObject);

Var

  r : real ;

Begin
  r := strtofloat ( edit1.text ) ;
  label1.Caption := inttostr ( trunc ( r ) ) ;

  label2.Caption := inttostr ( floor ( r ) ) ;

  label3.Caption := inttostr ( round ( r ) ) ;

  label4.Caption := inttostr ( ceil ( r ) ) ;

  label5.Caption := floattostr ( int ( r ) ) ;

End;
مثال 12-3

برنامه​اي بنويسيد كه حقوق و مزاياي يك كارمند را محاسبه و فيش حقوقي وي را در يك Memo  چاپ كند (شكل 12-5). (ماليات و بيمه نيز محاسبه شود)

در اين برنامه از اطلاعات زير راجع به هر كارمند استفاده مي​كنيم:

                                =     حقوق ناخالص

+      (ساعات رسمي × حقوق هر ساعت)

-       (ساعات اضافه كار × 4/1 برابر حقوق هر ساعت)

         (ساعات كسر كار × 2/1 برابر حقوق هر ساعت)

نرخ بيمه × حقوق ناخالص = بيمه

حقوق ناخالص قبلي + حق مسكن = حقوق ناخالص

محاسبه ماليات:

· از 0 تا 65000 تومان معاف از ماليات

· از 65000 تا 80000 تومان، سه درصد حقوق ناخالص
· از 80000 تا 100000 تومان، هفت درصد حقوق ناخالص
· از 100000 تا 200000 تومان، ده درصد حقوق ناخالص
· از 200000 تومان به بالا، بيست درصد حقوق ناخالص

حق اياب و ذهاب + ماليات - بيمه - حقوق ناخالص = حقوق خالص

Procedure TForm1.BitBtn1Click(SEnder: TObject);

Const

  bimeh_max_val = 110000 ;

  bimeh_rate    = 0.28 ;

Var

  //based variables
  base,

  housepay,

  travelpay,

  formalhour : integer ;

  //work data vars

  workedhour : integer ;

  //result variables

  puresalary,

  impuresalary,

  tax,

  bimeh      : integer ;

  //temporary vars

  m,n,

  payperhour : real ;
Begin
  // get data

  base       := strtoint ( edit3.text ) ;

  housepay   := strtoint ( edit4.text ) ;

  travelpay  := strtoint ( edit6.text ) ;

  formalhour := strtoint ( edit5.text ) ;

  workedhour := strtoint ( edit7.text ) ;

  // compute impuresallary

  payperhour := base / formalhour ;    

  m := payperhour * workedhour ; 
  if workedhour > formalhour then

    m := m + (( workedhour - formalhour ) * 0.4 * payperhour ) ;

  if workedhour < formalhour then

    m := m - (( formalhour - workedhour ) * 0.2 * payperhour ) ;

  impuresalary := trunc ( m ) ;

  //compute bimeh

  n := m ;

  if n >  bimeh_max_val then

     n := bimeh_max_val ;

  n := n * bimeh_rate ;

  bimeh := trunc ( n ) ;
// add housepay

  m := m + housepay ;

  // compute tax

  if      m <  65000  then

    n := 0

  else if m <  80000  then

    n := m * 0.03

  else if m <  100000 then

    n := m * 0.07

  else if m <  200000 then

    n := m * 0.10

  else if m >= 200000 then

    n := m * 0.20 ;

  tax := trunc ( n ) ;

  // perform decrimentation of bimeh and tax

  m := m - tax - bimeh ;

  // add travel pay

  m := m + travelpay ;

  puresalary := trunc ( m ) ;

  // showdata data

  statictext1.Caption := inttostr ( impuresalary ) ;

  statictext2.Caption := inttostr ( tax ) ;

  statictext3.Caption := inttostr ( puresalary ) ;

  statictext4.Caption := inttostr ( bimeh ) ;
  //print data in memo

  memo1.Lines.add ( '**********فيش حقوقي***********' ) ;

  memo1.Lines.add ( edit2.text+' '+edit1.text+'خانم/آقاي' ) ;

  memo1.Lines.add ( inttostr ( base ) +': پايه حقوق' ) ;

  memo1.Lines.add ( inttostr ( formalhour ) +': تعداد ساعات رسمي' ) ;

  memo1.Lines.add ( inttostr ( travelpay ) +': حق اياب و ذهاب  '+

                    inttostr ( housepay ) +':حق مسكن' ) ;

  memo1.Lines.add ( '' ) ;

  memo1.Lines.add ( inttostr ( workedhour ) +': ساعت كار ماه قبل' ) ;

  memo1.Lines.add ( inttostr ( impuresalary ) +': حقوق ناخالص' ) ;

  memo1.Lines.add ( inttostr ( tax ) +': ماليات' ) ;

  memo1.Lines.add ( inttostr ( bimeh ) +': بيمه' ) ;

  memo1.Lines.add ( '------------------' ) ;

  memo1.Lines.add ( inttostr ( puresalary ) +': حقوق خالص' ) ;

  memo1.Lines.add ( '' ) ;

  memo1.Lines.add ( '' ) ;

  memo1.Lines.add ( '-------------------------------------------------' ) ;

  memo1.Lines.add ( '' ) ;

End;
12-7- خطاها

در صورتيكه در برنامه خطا يا اشتباهي رخ دهد، در صورتيكه كامپايلر متوجه اين خطا شود، آن را به برنامه​نويس متذكر مي​شود. انواع خطاها در برنامه​نويسي :

الف) خطاهايي كه سبب توقف برنامه يا كامپايل آن مي​شوند. اين خطاها خود بر دو نوعند:

1. خطاي زمان كامپايل: دو نوع خطا در زمان كامپايل رخ مي​دهد:

خطاي نحوي يا Syntax error : اين گونه خطاها معمولا در اثر رعايت نكردن قوانين ساختاري زبان صورت مي​گيرد. به عنوان مثال با نگذاشتن سمي​كالن(;) بعد از يك دستور، چنين خطايي رخ مي​دهد. و يا به كارگيري متغيري كه در برنامه تعريف نشده است كه زبان از درك اين متغير ناتوان است.
خطاي معنايي يا Semantic error : اين خطا معمولا در اثر بكارگيري اشتباه يك دستور و يا انجام يك عمل غير مجاز در برنامه رخ مي​دهد. مثلا ريختن يك متغير از نوع اعشاري در يك متغير از نوع صحيح منجر به ايجاد خطاي معنايي مي​گردد.
2. خطاي زمان اجرا
: در اثر وجود خطاهاي زمان اجرا، برنامه كامپايل مي​شود، ولي در زمان اجرا به هنگام انجام خطا، متوقف مي‌گردد. مانند عمل تقسيم بر صفر كه كامپايلر از اجراي آن ناتوان است و هرگاه برنامه به چنين عملي رسيد، برنامه متوقف مي‌شود. هرگاه چنين خطايي در برنامه اتفاق افتاد و برنامه با يك پيغام خطا از كار ايستاد، براي متوقف كردن آن، از منوي Run، ‌بخش Program Reset را اجرا كنيد. با اين كار برنامه از حالت اجرا خارج مي​شود و در اين هنگام مي​توانيد نسبت به رفع خطا اقدام كنيد و بعد برنامه را مجدد اجرا نماييد. به عنوان مثالي ديگر از خطاي زمان اجرا، فراخواني فايلي از روي ديسك كه نشاني آن در برنامه داده شده است ولي چنين فايلي موجود نيست. در چنين حالتي نيز برنامه متوقف مي​گردد. البته در بعضي مواقع كه كامپايلر به وجود چنين خطاهاي پي ببرد، به طور خودكار در زمان كامپايل، اشاره
 و يا هشدار
 مي​دهد.

ب) خطاهايي كه سبب توقف برنامه و كامپايل نمي​شوند:
در مواقعي ممكن است كه برنامه كار كند ولي عملكرد درستي نداشته باشد. به عبارتي برنامه كار كند ولي جواب درستي ندهد. اينگونه از خطاها كه معمولا به خطاهاي منطقي مشهورند، از بكارگيري يك الگوريتم اشتباه ناشي مي‌شوند. برنامه نويس، الگوريتم را به درستي طراحي و پياده سازي نكرده است. در چنين حالاتي ممكن است قسمتي از برنامه اشتباه كند و نتيجه اشتباهي را بروز دهد. 
توجه داشته باشيد كه پيدا كردن خطاهاي غير متوقف كننده و مفهومي خيلي دشوار و ثقيل تر از ديگر خطاها است. چرا كه مكان خطاهاي متوقف كننده معلوم و خطاهاي غير متوقف كننده، غيرمعلوم است. شما مي​دانيد كه برنامه درست كار نمي​كند (و البته گاهي اين را هم نمي​دانيد!)، ولي نمي​دانيد كه كجاي برنامه مشكل دارد. در اينگونه مواقع براي رسيدن به قسمت نادرست برنامه مجبور به رديابي برنامه و بعضا اجراي تكه به تكه آن
 هستيم. براي اين كار ابتدا بايد برنامه را با كليدهاي  Ctrlو F9 كامپايل نمود و سپس با كليد F7 و F8، به صورت مرحله به مرحله اجرا نمود. كامپايلر روي خط در حال اجرا با خط آبي مي​ايستد. به اينكار دنبال
 كردن اجراي برنامه مي​گوئيم.

12-7-1- مراحل از بين بردن خطا

از بين بردن خطا مشكل​ترين قسمت كار برنامه​نويسي است. بنابراين همين كار بايد مبتني بر اصول و مراحل مشخصي انجام شود. به عنوان يك رويكرد عمومي، شكل 12-6 مراحل يافتن و از بين بردن خطا را نشان مي​دهد.

با انجام اين چهار مرحله مي​توان خطاهاي بوجود آمده در برنامه را برطرف نمود. اما هميشه پيدا كردن خطا كار ساده​اي نيست. خصوصا در مورد خطاهاي غير متوقف كننده و مفهومي كه در برنامه​هاي بزرگ رخ مي‌دهد.

12-7-2- اهميت خطاهاي كوچك

بر خلاف تصور عامه،‌ عوارض ناشي از خطاهاي كوچك، در نرم​افزارهاي رايانه​اي به همان اندازه كوچك نيستند. برخي اوقات خطاهاي بسيار كوچك، منجر به مختل شدن عملكرد كل يك سيستم نرم‌افزاري عظيم مي‌شود. براي هيچكس قابل تصور نيست كه اگر درب كوچك يك انبار در يكي از طبقات يك آسمانخراش بسيار بزرگ، محكم به هم بخورد و آسيب ببيند، ناگهان كل آسمانخراش فرو بريزد! اما چنين چيزي در سيستمهاي نرم​افزاري ممكن است. مثالهاي متعددي از شكستهاي عظيم در سيستمهاي رايانه​اي، بر اثر خطاهاي كوچك وجود دارد. به عنوان مثال سفينه مارينر كه توسط سازمان فضايي ناسا به هوا پرتاپ شد، در ابتداي صعود منفجر شد. بررسيهاي انجام شده بعدي نشان داد كه علت اين خرابي ناشي از عملكرد نادرست نرم‌افزار كنترل كننده اين سفينه بود. در هنگام انتقال نرم‌افزار به رايانه سفينه تنها يك كاراكتر خط تيره،‌ در برنامه چند صد هزارخطي آن جا افتاده بود.  همين موضوع باعث شد كه كل برنامه در يك حلقه بي​پايان گرفتار شده و  همين امر موجب انفجار سفينه شد. اين مثال و مثالهاي مشابه آن، حساسيت كار برنامه​نويسي را نشان مي​دهد.
اهميت آزمايش و بررسي صحيح يك برنامه تا حدي است كه بيش از 50 درصد از فعاليتهاي مربوط به ايجاد يك برنامه، به آزمايش و يافتن خطاهاي آن تعلق دارد. و البته هر چقدر كه برنامه​نويسي اصولي و با رعايت نظم و قواعد صحيح برنامه​نويسي انجام شود، اين زمان كمتر خواهد بود.

12-8- ثابت​ها

فرض كنيد در برنامه حقوق پرسنل كه آن را در قسمت قبل نوشتيم، لازم است تغييري در جدول مالياتي پديد بياوريد. در اين صورت بايد در تمام برنامه هر جا كه از اين جدول استفاده شده است، تغيير انجام شود. اما اين كار بسيار مشكل است. در چنين حالتي كه در داخل برنامه دائما مقدار ثابتي مورد استفاده قرار مي​گيرد، مي​توان از يك ثابت استفاده كرد. ثابت، كميتي است كه هنگام اجراي برنامه تغيير نمي‌كند. نحوه تعريف ثابتها به صورت زير است:

ابتدا كلمهConst  و سپس نام ثابتها به همراه مقدارشان :

Const 

    rate = 2.3 ;

    yes  = True ;

    no   = False ;

    wait_msg = ‘please wait while processing’ ;
نوع اطلاعاتي كه در يك ثابت قرار مي​گيرد، نوع ثابت را مشخص مي​كند. مثلا اگر ثابتي تعريف شود و در آن يك عدد اعشاري قرار گيرد، ثابت اعشاري و اگر در آن عدد صحيح قرار گيرد، ثابت از نوع صحيح خواهد بود. استفاده از ثابتها در برنامه مزاياي زيادي دارد. مهمترين مزيت آن است كه در صورت تغيير، اعمال تغييرات ساده​تر است. فرض كنيد در برنامه​اي ثابتي تعريف كرديد و از آن به تعداد زيادي در برنامه تكرار كرديد. در صورتيكه لازم به تغييري در آن باشد، لازم نيست تا يكي يكي به نقاط استفاده شده در آن برويد و مقدار آن را عوض كنيد. بلكه تنها مقداري كه در تعريف براي ثابت در نظرگرفته بوديد را عوض مي​كنيد. 
مزيت بعدي استفاده از ثابتها خوانايي بيشتر برنامه است. مثلا به جاي تكرار عدد در يك فرمول، نام آن قرار داده مي​شود. به عنوان مثال ممكن است هم مقدار درصد ماليات و هم مقدار ضريب اضافه كار،‌ هر دو 0.25 باشد. در صورتي كه مستقيما از عدد در فرمول استفاده شود، معلوم نيست كه اين 0.25 مربوط به كدام يك از اين دو است. اما اگر دو ثابت با دو نام متقاوت اما مقدار يكسان 0.25 تعريف شود،‌ بكارگيري اين ثابتها در فرمول، موضوع و چگونگي محاسبه را روشن تر مي‌كند. 

به عنوان مزيت ديگر، اشتباه شدن در تايپ نام ثابت،‌ كمتر از اشتباه شدن در نوشتن ارقام يك عدد اتفاق مي افتد. زيرا كامپايلر نام اشتباه را به سرعت تشخيص داده و به ما اعلام مي​كند، اما شماره اشتباه كه مثلا يك صفر زيادتر داشته باشد، توسط كامپايلر قابل تشخيص نيست.

تمرين

برنامه حقوق پرسنل را با ثابتها بازنويسي كنيد.
نكته- با توجه به تعريف ثابت، مي​دانيم كه امكان تغيير آن در برنامه وجود ندارد و تنها همان مقداري كه در ابتدا به آن نسبت داده شد است، در برنامه به كار مي​رود. اما در حالاتي كه بخواهيم ثابت اين خاصيت را داشته باشد كه بتواند تغيير كند، همانند تعريف متغير نوع آن را مشخص كرده و سپس مقداري را به آن نسبت مي​دهيم. مثلا:

Const

                          I : integer = ‘25’ ;
با انجام چنين تعريفي، ثابت، هم به معناي ثابت است و مقدار اوليه اي دارد، و هم به عنوان متغير در برنامه اجازه تغيير آن را داريم. به عبارت ديگر اينگونه تعريف ثابت در يك برنامه، تعريف يك متغير با مقدار اوليه مشخص است. 
مثال 12-4

برنامه​اي بنويسيد كه يك عدد n (بين صفر تا بيست) را دريافت كند و فرمول زير را محاسبه نمايد:

n^2+(n-1)+(n-2)^2+(n-3)+ …. + 0
براي تشخيص اينكه اين عدد بايد به توان دو برسد يا خير، اين برنامه را از دو راه زير بنويسيد:

راه اول) از تابع odd و even استفاده كنيد.

راه دوم) از يك متغير مانند f استفاده كنيد و آن را بار اول يك كنيد. سپس هر بار كه يك باشد، آن را صفر و هر بار كه صفر باشد، آن را يك كنيد. ( به چنين متغيري متغير پرچم يا flag گفته مي​شود) فرم اين برنامه در شكل 12-7 آمده است.


Procedure TForm1.Button1Click(SEnder: TObject);

var

  n,

  i,

  sum : integer ;

Begin
  n := strtoint ( edit1.text ) ;

  sum := 0 ;

  for i := 1 to n do

    Begin
      if not odd ( i ) then

        sum := sum + ( i * i )

      else

        sum := sum + i ;

    End ;

  statictext1.caption := inttostr ( sum ) ;

End;
Procedure TForm1.Button2Click(SEnder: TObject);

var

  n,

  i,

  sum  : integer ;

  f : byte ;

Begin
  n := strtoint ( edit1.text ) ;

  sum := 0 ;

  f := 0 ;

  for i := 1 to n do

    Begin
      if f = 1 then

        Begin sum := sum + ( i * i ) ; f := 0 ; End
      else

        Begin sum := sum + i ; f := 1 ; End ;

    End ;

  statictext2.caption := inttostr ( sum ) ;

End;
12-9- متغيرهاي منطقي

در روش دوم در برنامه گذشته، متغير پرچم، ما را به ياد مخابره پيام با استفاده از پرچم مي​اندازد. چنين متغيري فقط دو حالت را به خود مي​گيرد. پرچم بالا يا پرچم پايين. درست يا غلط. صفر يا يك. اما مي​دانيم كه متغير f در مثال قبلي كه يك byte بود، مي​تواند مقادير ديگري نيز به خود بگيرد. حال اين سئوال پش مي​آيد كه اگر مثلا اشتباها مقدار 2 در آن قرار دهيم، چه اتفاقي مي​افتد؟
براي اينكه چنين مشكلي بوجود نيايد، از متغيري استفاده مي​كنيم كه تنها دو مقدار "درست" يا "غلط" را به خود مي​گيرد. چنين نوع متغيري Boolean ناميده مي​شود و مقادير True و False را به خود مي​گيرد. 

                                              Var           

b : Boolean ;

…

b := True ;

…

b := False ;
براي كنترل يك متغير منطقي به عنوان يك شرط، لازم نيست تا آن را با True يا False مقايسه كنيم. تنها كافي است همان متغير را به عنوان شرط بنويسيم. مثلا

if  b then 

  …. ;

مانند آن است كه بنويسيم:

                                                                 if  b = True then 

 …. ;
همچنين براي معكوس كردن يك متغير Boolean آن را با خودش not مي​كنيم. مثلا:

b := not b ;
با اين كار مقدار متغير b به عكس خودش تبديل مي​شود. اگر True باشد، False و اگر False باشد، True مي​شود.

12-9-1 انواع Boolean ها

هر چند كه يك Boolean فقط مقدار صفر يا يك را مي​گيرد، ولي در كوچكترين بلاك حافظه، يعني بايت قرار مي​گيرد. اما برخي اوقات براي ساده​شدن عمليات لازم است تا اندازه يك متغير Boolean بيش از يك بايت باشد. هر چند كه در عمل آن تاثيري نخواهد گذاشت. اين انواع عبارتند از:

                          (Boolean همان ) bytebool , wordbool , longbool
مثال 12-5

با استفاده از اين نوع متغيرهاي بولين، مثال 12-4 را بازنويسي مي​كنيم. 

Procedure TForm1.Button3Click(SEnder: TObject);

var

   n,

   i,

   sum  : integer ;

   even : Boolean ;

Begin
  n := strtoint ( edit1.text ) ;

  sum := 0 ;

  even := False ;

  for i := 1 to n do

    Begin
      if even then

        sum := sum + ( i * i )

      else

        sum := sum + i ;

      even := not even ;

    End ;

  statictext1.caption := inttostr ( sum ) ;

End;
همانطوري كه مشاهده مي​كنيد، علاوه بر ساده​شدن برنامه و افزايش خوانايي آن، امكان اشتباه گفته شده نيز رفع مي​گردد.
12-10- عملگرها يا اپراتورها

يك عملگر يا اپراتور، همانطور كه از نام آن پيداست، عملي را بر يك يا دو عنصر انجام مي​دهد. مثلا عملگر "+" دو عنصر چپ و راست خود را جمع مي​كند. اغلب عملگرها دو عملوند
 (يا اپرند) دارند. به اين عملگرها باينري
 گفته مي​شود، مانند 2+3. برخي از عملگرها نيز يك عملوند دارند. به اين عملگرها unary گفته مي​شود. مانند عملگر منفي.

12-10-1- عملگرهاي رياضي

 اين عملگرها اعمال محاسبات را روي عملوندها انجام مي​دهند. انواع عملگرهاي رياضي در جدول 3-12 آمده​اند:

	مثال
	نوع نتيجه
	نوع اپرند
	نوع عمليات
	عملگر

	 X – Y
	Integer , real
	Integer , real
	جمع
	+

	Result – 1
	Integer , real
	Integer , real
	تفريق
	-

	P * InterestRate

	Integer , real
	Integer , real
	ضرب
	*

	X / 2
	real
	real
	تقسيم اعشاري
	/

	Total DIV unitsize
	integer
	integer
	تقسيم صحيح
	IV

	Y MOD 6
	integer
	Integer
	باقي مانده
	MOD

	+7
	Integer , real
	Integer , real
	مثبت
	+ (unary)

	-x
	Integer , real
	Integer , real
	منفي
	- (unary)


12-10-2- عملگرهاي مقايسه

اين عملگرها رابطه بين عملوندها را مشخص مي​كنند. انواع عملگرهاي رياضي در جدول 4-12 آمده​اند:

	مثال
	نوع نتيجه
	نوع اپرند
	نوع عمليات
	عملگر

	I = max
	Boolean
	Simple , string
	مساوي
	=

	X <> y
	Boolean
	Simple , string
	مخالف
	<>

	X < y
	Boolean
	Simple  , string , PChar
	كوچكتر از
	<

	Len > 0
	Boolean
	Simple , string , PChat
	بزرگتر از
	>

	Cnt <= 1 
	Boolean
	Simple , string , PChar
	كوچكتر مساوي
	<=

	I >= 1
	Boolean
	Simple , string , PChar
	بزرگتر مساوي
	>=


تذكر: خروجي يك عبارت مقايسه​اي، يك مقدار Boolean است.

a < 2 ( True
k > 3 ( False
b = True ( True
آيا اكنون دليل نكته​اي كه در مورد عدم نياز مقايسه متغير منطقي در عبارات شرطي گفته شده بود، پي برديد؟

12-10-3- اپراتورهاي منطقي

عملگرهاي منطقي بر روي عبارات منطقي عمل مي​كنند. اين عملگرها داراي دو ارزش درست و نادرست هستند. به عنوان مثال اگر حاصل عبارت منطقي 2a> درست باشد، مقدار True برگردانده مي‌شود و اگر درست نباشد، False برگردانده مي​شود. از اين رو تمامي عبارتهاي شرطي، نتيجه منطقي دارند. مثلا نتيجه عبارت then if b<3، يك ارزش منطقي  Trueيا False است. انواع عملگرهاي منطقي در جدول 5-12 آمده​اند.


	مثال
	نوع نتيجه
	نوع اپرند
	نوع عمليات
	عملگر

	Not ( c in myset)
	Boolean
	Boolean
	نقيض

	not

	Done and (total>0)
	Boolean
	Boolean
	و

	And

	A or B
	Boolean
	Boolean
	يا

	or

	A xor B
	Boolean
	Boolean
	ياي انحصاري

	xor


عملگر not داراي يك عملوند است و عملوند را نقيض مي​كند، يعني اگر عملوند True باشد، False و اگر False باشد، True مي​كند. عملگر and داراي دو عملوند است. اگر هر دو عملوند True باشند، نتيجه True، و در غير اين صورت، نتيجه False است. عملگر or نيز دو عملوندي است و تنها زماني كه هر دو عملوند False باشند، نتيجه False است. در غير اين صورت True است. عملگر xor هم دو عملوندي و تنها در حالاتي كه هر دو عملوند ارزش متفاوتي را داشته باشند، نتيجه True خواهد بود.


البته عملوندهاي هر يك از عملگرها خود مي​تواند شامل يك عبارت باشد. در اين صورت، منظور از ارزش عملوند، ارزش عبارت خواهد بود. 

12-11- تقدم عمليات عملگرها

هنگامي كه در عبارتي چندين عملگر وجود داشته باشد، اين سئوال پيش مي​آيد كه ترتيب اجراي هر يك از عملگرها چگونه خواهد بود؟ به عنوان مثال در عبارت A/B+C، ابتدا تقسيم انجام مي​شود يا جمع؟ براي پاسخ به اين سئوالات بايد تقدم عملگرها را مشخص كرد.

ترتيب اجراي عملگرها در يك عبارت، از چپ به راست، و بر حسب تقدم و اولويتي كه در جدول 6-12 آمده است، انجام مي​شود.
	تقدم
	عملگرها

	بالاترين تقدم
	@ , not , -(unary) , +(unary)

	تقدم دوم
	* , / , div , mod , and , shl , ahr , as

	تقدم سوم
	+(binary) , -(binary) , or , xor

	پائين​ترين تقدم
	= , <> , < , > , <= , >= , in , is


12-12- زيربرنامه

وقتي يك ليست بزرگ از اسامي داشته باشيد (مثلا اسامي دانشجويان يك دانشگاه)، براي نگهداري آن چه مي​كنيد؟ قطعا آن را به بخشهاي كوچكتري تقسيم كرده و براي هر يك از آن بخشها، يك عنوان قرار مي​دهيد. مثلا دانشجويان دانشكده فني و ... و يا حتي ليست يك دانشكده را نيز به ليست گروه​ها و ليست ورودي​هاي سالهاي مختلف تفكيك مي​كنيد. با اين كار ليست كل دانشجويان به زيرليست​هاي كوچكتري تقسيم شده است كه هر يك از اين ليست​ها مجموعه مستقلي را مشخص مي​كند. اين كار باعث سهولت در نگهداري ليست دانشجويان و درك بهتري از ليست​ها مي​شود.
وقتي كه ليستي از كارهائي كه بايد انجام دهيد، داشته باشيد، چطور. خوب آنرا هم به ليستهاي كوچكتر تفكيك مي​كنيم، به صورتي كه هر يك از ليست​ها يك كار مستقل را مشخص كند. 
در ساختمان برنامه​ها نيز چنين كاري انجام مي​دهيم. برنامه​هايي كه تا كنون نوشتيم، برنامه​هاي ساده و كوچكي بودند. اما ممكن است با مسائلي برخورد كنيم كه برنامه آنها بسيار طولاني و به چند صد سطر برسد. در اين صورت بهتر است تا مسئله را به بخشهاي كوچكتري تقسيم كنيم، به طوريكه هر بخش، كار خاصي را انجام دهد و سپس براي هر بخش، برنامه مربوط به آن را بنويسيم. در اين صورت برنامه بزرگ ما به چندين برنامه كوچك تفكيك مي​شود كه به هر يك از اين تكه برنامه​ها يك زيربرنامه گوييم.
معيار اصلي تفكيك برنامه به زيربرنامه​ها، كارهايي است كه انجام مي​شود (و نه اندازه آن). بنابراين هر كار مستقل در يك زيربرنامه قرار مي​گيرد. زيربرنامه​ها را به دو صورت رويه
 و تابع
 پياده​سازي مي‌كنند كه در ادامه با هر يك از اين دو نوع زيربرنامه آشنا خواهيم شد.
12-13- رويه

براي نوشتن يك رويه، ابتدا در سطر اول عبارت Procedure و سپس نام رويه، و در داخل پرانتز ليست پارامترهاي رويه را مشخص مي​كنيم، يعني به صورت زير:

Procedure نام زيربرنامه   (ليست پارامترها)  ;
پارامترها، همان مقاديري هستند كه از برنامه اصلي به رويه انتقال پيدا مي​كنند و رويه روي اين مقادير، عمليات خاص خود را انجام مي​دهد. در واقع پارامترها، عامل ارتباط برنامه و رويه هستند. پس از اعلان رويه، عمليات داخل رويه را مي​نويسيم. اين عمليات همان كاري است كه رويه بايد انجام دهد. عمليات داخل رويه همانند يك برنامه، مابين Begin و End نوشته مي​شود.

مثال 12-6

رويه​اي كه دو عدد را گرفته و با هم جمع مي​كند:
Procedure Sum (X,Y : integer) ;

Var

  Result_sum : integer ;

Begin
  Result_sum := x +  y ;

End ;
متغيرهاي x و y كه در داخل پرانتز قرار گرفته​اند، همان پارامترهايي هستند كه زيربرنامه Sum ارسال شده​اند. اين دو متغير در اين زيرروال با هم جمع شده و در متغيري با نام Result_sum قرار مي‌گيرند.
تذكر: اين زيربرنامه عمل جمع را انجام مي​دهد، ولي نتيجه آن چه مي​شود؟ پاسخ آن را كمي جلوتر خواهيم ديد.
به عنوان مثال به زيربرنامه زير توجه كنيد. اين زيربرنامه چه كاري را انجام مي​دهد:

Procedure  NumString(N: Integer;  var  S:  string );

 var

    V: Integer;

  Begin
    V := Abs(N);

    S := '';

     repeat

      S := Chr(V  mod  10 + Ord('0')) + S;

      V := V  div  10;

     until  V = 0;

     if  N < 0  then  S := '-' + S;

  End ;
12-13-1- بكارگيري رويه

پس از تعريف يك رويه، براي استفاده بايد آن را در برنامه اصلي صدا زد. براي اين كار نام رويه را به همراه پارامترهاي آن صدا مي​زنيم. توجه داشته باشيد كه نوع پارامترهاي ارسال شده به رويه بايد با نوع پارامترهايي كه در تعريف رويه آمده​اند، يكي باشد. پارامترهايي كه در قسمت تعريف رويه مي​آ​يند، پارامترهاي مجازي و پارامترهايي كه در فراخواني رويه مي​آيند، پارامترهاي واقعي نام دارند. به عنوان مثال براي استفاده از رويه مثال بالا به صورت زير عمل مي​كنيم:
Sum(5,6) ;
با اين دستور مقدار 5 در x و مقدار 6 در متغير Y به عنوان دو پارامتر به رويه Sum ارسال مي​گردند و اين رويه، كار خود را بر روي اين دو متغير انجام مي​دهد.

12-13-2- محل قرارگرفتن رويه

رويه​ها بايد قبل از محل بكارگيري​شان قرار گيرند. به عبارتي اگر در يك برنامه بخواهيم از يك رويه استفاده كنيم و آن را صدا بزنيم، بايد حتما آن رويه قبل از اين برنامه تعريف شده باشد، نه بعد از آن. رويه​اي كه در قسمت Implementation يك واحد تعريف مي​شود، فقط در همان واحد قابل دسترسي است. البته مي​توان كاري كرد كه روال در ديگر واحدها نيز قابل دسترسي باشند. براي اينكار بايستي عنوان رويه را در بخش Interface واحد اضافه نمود و واحد ديگري را كه بخواهد از رويه اين واحد استفاده كند را از منوي File، Use Unit كرد. در فصل پانزدهم، راجع به چگونگي اين كار توضيحات بيشتري داده شده است.

نكته

نتيجه محاسبه جمع در مثال 12-6 در بيرون آن زيربرنامه و در هنگام استفاده از آن قابل دسترسي نيست. بايد به نحوي اين نتيجه را به خارج از برنامه منتقل نمود. يك راه اين كار استفاده از متغيرهاي جهاني (global) است. چنين كاري معمولا پيشنهاد نمي​شود. يك راه بهتر آن است كه از پارامترها براي اين كار استفاده شود. رويه​ها با گرفتن چند پارامتر (و يا حتي بدون گرفتن پارامتر) اعمالي ر انجام مي​دهند. در صورتي كه بخواهيم پس از انجام عمليات در رويه، مقداري به عنوان نتيجه به برنامه اصلي (فراخواننده رويه) بازگشت كند، بايد در ابتداي پارامتري كه قرار است به عنوان نتيجه برگردانده​شود عبارت var را بنويسيم. به مثال 12-7 توجه شود.

مثال 12-7

رويه​اي بنويسيد كه دو عدد صحيح را بگيرد و اولي را به توان دومي برساند و حاصل را به برنامه برگرداند.

براي اين كار در رويه​اي با نام Power هم دو عدد را به عنوان پارامتر ورودي در نظر مي​گيريم، و هم متغيري با نامه Result را به عنوان پارامتر سوم و براي برگشت دادن نتيجه معرفي مي​كنيم.
Procedure power ( n,m : integer ; var result : int64 ) ;

var

  i : integer ;

Begin
  result := 1 ;

  for i := 1 to m do

    result := result * n ;

End ;

Procedure TForm1.Button1Click(SEnder: TObject);

var

  k,h : integer ;

  r   : int64 ;

Begin
  k := strtoint ( edit1.text ) ;

  h := strtoint ( edit2.text ) ;

  power ( k,h, r );

  statictext1.Caption := inttostr ( r ) ;

End;
تمرين

مثال 12-6 را با توجه به اين موضوع مجددا بازنويسي كنيد.

پس از اجراي برنامه زير و فراخواني رويه، مقادير خروجي چه اعدادي خواهند بود؟

Procedure x ( n : integer ; var m : integer ) ;

  Begin
    n := 10 ;

    m := 10 ;

  End ;

Procedure TForm1.Button2Click(SEnder: TObject);

var

  k ,h : integer ;

Begin
  k := 5 ;

  h := 5 ;

  x ( k , h ) ;

  statictext2.caption := inttostr ( k ) ;

  statictext3.caption := inttostr ( h ) ;

End;
نكته

اگر يك در رويه بخواهيم از اشياء داخل فرم مانند كليد و برچسب و غيره استفاده كنيم، بايد قبل از نام رويه، كلمه Tform1. را بنويسيم و اعلان رويه را در بخش Type اضافه نماييم. به عنوان مثال فرض كنيد در يك رويه مي​خواهيم به عنوان يك برچسب در داخل فرم، عبارتي نسبت دهيم. كد زير چگونگي انجام اين كار را نشان مي​دهد:

....
Type

  TForm1 = class(TForm)

    Button1: TButton;

    Label1: TLabel;

    Procedure Button1Click(SEnder: TObject);

    Procedure p ;

  private

    { Private declarations }

  public

    { Public declarations }

  End;

Var

  Form1: TForm1;

implementation

{$R *.dfm}

Procedure TForm1.p ;

Begin
  label1.caption := 'reza' ;

End ;
12-14- تابع

برخي مواقع يك زيربرنامه فقط بايد يك مقدار برگرداند. در چنين شرايطي از يك تابع استفاده مي​شود. يك تابع يك مقدار را بر مي​گرداند. شيوه نوشتن يك تابع همانند
 يك رويه است، منتهي با اين تفاوت كه ابتدا نام Function ذكر مي​شود و پس از تعريف پارامترها، با گذاشتن ":" نوع خروجي تابع ذكر مي​شود:

Function نام تابع   (ليست پارامترها)  نوع خروجي :

مثال 12-8

نمونه​اي از يك تابع را در زير مشاهده مي​كنيد. اين تابع دو پارامتر، يكي از نوع حقيقي و ديگري از نوع صحيح مي​گيرد و يك عدد حقيقي برمي​گرداند. همانطوري​كه از عملكرد اين تابع مشاهده مي​شود، اين تابع عمل توان را انجام مي​دهد. بنابراين بار ارسال دو مقدار به اين تابع مي​توان عمليات به توان رساندن را در برنامه انجام داد.

Function  Power(X: Real; Y: Integer): Real;

 Var

    I: Integer;

  Begin
    Result := 1.0;

    I := Y;

     while  I > 0  do

     Begin
       if  Odd(I)  then  Result := Result * X;

      I := I  div  2;

      X := Sqr(X);

     End ;

  End ;
مقدار خروجي تابع، در متغير result قرار مي​گيرد. اين متغير به طور پيش فرض در توابع تعريف شده است و براي برگرداندن مقدار خروجي به كار مي​رود. بنابراين در مثال فوق مقداري كه پس از انجام عمليات تابع در result قرار مي​گيرد، همان خروجي تابع است. البته براي برگرداندن خروجي تابع مي​توان از نام خود تابع نيز استفاده كرد. مثلا در مثال 12-8 به جاي به جاي كلمه Result، مي​توان كلمه Power را نوشت.

مثال 12-9

برنامه​اي بنويسيد كه عددي را از ورودي گرفته و تشخيص دهد كه آيا اول است يا خير.

عدد اول، عددي است كه بر هيچ عدد كوچكتر از نصف خودش بجز 1 بخش​پذير نباشد. در تابع نوشته شده، ابتدا فرض شده است كه عدد اول است.

Function prime (const x:intger) : Boolean ;

Var

  J : integer ;

  Prime1 : Boolean ;

Begin
  Prime1 := True ;

  J := 2 ;

  While (j<=x div 2) and (prime1) do

    Begin
      If x mod j=0 then prime1 := False ;

      J := j + 1 ;

    End 

    Prime := prime1 ;

End ;
Procedure TForm1.Button1Click(SEnder: TObject);

Var  number : intger ;

Begin
  Number := strtoint (edit1.text) ;

If prime(number) then

  Label1.caption := ‘number is primary’

Else

  Label1.caption := ‘number is not primary’ ;

End ;
12-14-1-  فرق بين رويه و تابع (Procedure و Function)

Procedureها مي​توانند چندين ورودي و چندين خروجي داشته باشند، ولي Functionها چند ورودي و تنها يك خروجي دارند. Procedureهايي كه يك خروجي دارند را مي​توان به صورت Function نوشت، چرا كه كاركردن با توابع معمولا آسان​تر است.

12-15- انواع پارامترها

Call by value و Call by reference-  سؤال مهمي در ارتباط با زيربرنامه​ها وجود دارد و آن اينكه آيا مقدار پارامترهاي ورودي به زيربرنامه در صورتي كه اعمالي روي آنها انجام گيرد، پس از خاتمه يافتن كار زيربرنامه تغيير مي​كند يا خير! به عنوان مثال آيا در برنامه زير، مقدار متغير X پس از اتمام كار زيربرنامه تغيير مي​كند يا همان مقدار 5 است؟
Procedure test(x : integer) ;

  Begin
    x := x * 2 ;

  End ;

Var

  x : integer ;

Begin  
  x  := 5 ;

Test (x) ;

// x = 5 or x = 10 ?

End ;
براي پاسخ به اين سئوال، به اين دو زيربرنامه توجه نماييد:

Procedure  DoubleByValue(X: Integer);     // X is a value parameter

 Begin
    X := X * 2;

  End ;

 Procedure  DoubleByRef( var  X: Integer);   // X is a variant parameter

 Begin
    X := X * 2;

 End ;
 همانطوري كه مشاهده مي​كنيد ظاهرا هر دو زيربرنامه يا روال، كار مشابهي را انجام مي​دهند. هر دو مقداري متغير ورودي را دو برابر مي​كنند. ولي يك تفاوت جزئي دارند. در روال دوم در هنگام تعريف پارامتر، عبارت var به كار رفته است. در صورتي كه در روال اول چنين نيست. همين اختلاف باعث تفاوت در عملكرد اين دو روال مي​شود. در روال اول، ارزش پارمتر ارسال شده پس از اتمام زيربرنامه تغييرنمي​كند و مقدار جديدي به خود نمي​گيرد. به فراخواني پارامتر اين زير برنامه،‌ يك فراخواني از نوع  call by valueو به پارامتر ارسالي يك value parameter مي‌گوييم (مي​توان چنين فرض كرد كه يك كپي از پارامتر اصلي به زيربرنامه ارسال مي​شود، نه خود پارامتر. بنابراين در بازگشت به برنامه اصلي، مجددا مقدار اوليه خود پارامتر حفظ مي​شود). ولي در روال دوم، ارزش پارامتر ارسال شده پس از اتمام زيربرنامه تغيير مي​كند. به اين فراخواني پارامتر اين روال يك فراخواني از نوع call by reference و به پارامتر ارسالي يك  variant parameter مي‌گوييم. بنابراين متغير ورودي به اين دو روال كه همان x است، پس از انجام عمليات در روال اول تغيير نمي​كند، در صورتي كه در روال دوم تغيير مي​كند. 

مثال 12-10

در برنامه زير كه دو رويه DoubleByValue و DoubleByRef در آن به كار گرفته شده​اند، مقدار label1، 4 ولي مقدار label2 ، 8 چاپ مي​شود.

Var

   I, J :Integer;

  Begin
    I := 4;

    J := 4;

    DoubleByValue(I);   

    DoubleByRef(J); 

    Label1.caption := inttostr(I) ;

    Label2.caption := inttostr(J) ;
  End ;
Constant- شبيه call by value است. منتهي فرقش در اين است كه در متن برنامه اجازه تغيير پارامتر داده نمي شود. البته اگر constant  هم نگذاريم مقدار اصلي پارامتر حفظ مي​شود. 

Function  CompareStr( const  S1, S2:  string ): Integer;

Out- پارامتر out، همانند يك variant parameter با ارجاع به زيربرنامه ارسال مي​شود. با بكارگيري پارامتر out ارزش اوليه متغير ارجاع داده شده پاك مي​شود.
Procedure  GetInfo( out  Info: SomeRecordType);

  Begin
  ...

  End ;

 Var  MyRecord: SomeRecordType;

 ...

  GetInfo(MyRecord);
استفاده از اين نوع پارامترها بيشتر براي اجتناب از اشتباه از خطاهاي برنامه​نويسي و كمك به برنامه​نويس براي نوشتن برنامه صحيح است. بنابراين بهتر است در شرايطي كه مي​دانيم پارامتر ورودي در متن برنامه ماهيتا نبايد تغيير كند، از پارامتر const و در صورتي كه مي‌دانيم داده خروجي به ورودي بستگي ندارد، از پارامتر out استفاده مي​كنيم.
12-16- دلايل استفاده از زيربرنامه​ها

3. با استفاده از زيربرنامه​ها، مي​توان اعمالي را كه در برنامه چندين بار تكرار مي​گردند، در يك زيربرنامه قرار داد و در مواقع نياز آن را صدا زد. با اين كار از يك زيربرنامه به دفعات مختلف مي​توان در برنامه اصلي استفاده كرد. بنابراين اعمال تكراري در برنامه تنها محدود به يك زيربرنامه شده و اين امر باعث سهولت در نوشتن برنامه مي​شود.
4. استفاده از زيربرنامه​ها خوانايي برنامه را زياد مي​كنند. اين افزايش خوانايي به دليل اين است كه با قرار گرفتن يك سري اعمال در زيربرنامه​ها، تنها اهداف كلي در قسمت اصلي برنامه قرار مي​گيرند و بنابراين با مشاهده قسمت اصلي برنامه، مي​توان به سرعت به هدف آن پي برد. زيربرنامه ما را از درگير شدن در جزئيات بيش از حد باز مي دارد و به ما كمك مي‌كند تا نگاه يكپارچه و منسجمي را به كل برنامه يا يك تكه از آن داشته باشيم.
5. ايجاد استقلال، دليل بعدي براي استفاده از زيربرنامه​ها است. استقلال براي يك برنامه بدين معني است كه با تجزيه شدن يك برنامه به تكه برنامه​هاي مجزا، هر تكه، كار مخصوص به خود را انجام مي​دهد و تاثير مستقيمي بر ديگر اجزاي برنامه ندارد. چرا كه هر تكه با قرار گرفتن در روالي جداگانه، از دسترسي و تاثير مستقيم ديگر تكه​ها در امان است. اين كار سبب نظم بيشتر و عدم تداخل در يك برنامه مي​شود.
6. دليل ديگر براي استفاده از زيربرنامه​ها، كوچك شدن حجم قسمت اصلي برنامه است. البته كوچكي بدين معنا نيست كه هر گاه با مشاهده يك برنامه بزرگ آن را شكسته و به برنامه​هاي كوچكي تقسيم كنيم. بلكه معيار اصلي براي شكستن يك برنامه و كوچك نمودن آن بر اساس كارها و عمليات مختلفي است كه در برنامه صورت مي​گيرد. بدين معني كه ببينيم چه عمليات مستقلي در برنامه وجود دارد و هر عمل را در يك زيربرنامه قرار دهيم.
7. نوشتن برنامه، آزمايش و تصحيح آن با بكارگيري زيربرنامه​ها ساده‌تر خواهد بود.
12-17- متغيرهاي محلي و غير محلي

وقتي يك متغير در يك زيربرنامه تعريف مي​شود، فقط در همان زيربرنامه قابل استفاده است. به عبارت ديگر حوزه دسترسي به آن متغير محدود به همان زيربرنامه مي​شود. بنابراين اگر جاي ديگري بخواهيم از آن استفاده كنيم، كامپايلر آن را نمي​شناسد. به اين متغير يك متغير محلي
 گفته مي​شود.
اما اگر بخواهيم متغيري را تعريف كنيم كه براي تمام قسمتهاي برنامه قابل استفاده باشد، بايد از متغيرهاي غيرمحلي يا مرجع
 استفاده كنيم. متغيرهاي مرجع در تمامي زيربرنامه​ها قابل استفاده هستند. البته غيرمحلي يا مرجع بودن خود نسبي است. زيرا با تعريف يك متغير مرجع، تنها براي قسمتهاي بعد از تعريف آن شناخته شده است. اگر بخواهيم يك متغير در كل يك واحد (Unit) عمومي باشد، آن را در ابتدي بخش Implementation تعريف مي​كنيم. در اين صورت هر زيربرنامه مي​تواند از اين متغير استفاده كند و اگر بخواهيم اين متغير در واحدهاي ديگر هم قابل استفاده باشد، آن را در بخش Interface تعريف مي​كنيم.
چند نكته

اگر متغيري از نوع مرجع تعريف شده باشد، و متغير ديگري نيز با همان نام در زيربرنامه​اي تعريف شده باشد، در اينصورت در زيربرنامه، تنها متغير محلي قابل دسترسي است. يعني هميشه متغير داخلي اولويت دارد. مگر آنكه قبل از نام متغير مرجع، نام واحد آن را نيز اضافه كنيم. مثلا:
Unit1.x
زيربرنامه محلي- يك زيربرنامه نيز مانند يك متغير، مي​تواند محلي باشد. يك زيربرنامه محلي فقط در زيربرنامه تعريف شده قابل دسترسي است. به عنوان مثال در برنامه زير، روال NestedProc، چون در يك زيربرنامه قرار گرفته است، تنها در همين زيربرنامه شناخته شده است و بقيه اجزاي برنامه اين روال را نمي​شناسند.
Procedure  DoSomething(S:  string );

   Var

      X, Y: Integer;

   Procedure  NestedProc(S:  string );

      Begin
        ...

      End ;
Begin
    ...

    NestedProc(S);

     ...

End ;

انتخاب نام�هاي مناسب و بامعنا، كمك زيادي به خوانايي و زيبايي برنامه مي�كند.

